IHOP Algebras, R-Elements, and WG

IHOP Algebras and R-Elements

Definition. An Involutive Hopf Algebra (IHOP, but only here) is a vector space H that has

- An algebra structure ($m\colon H\otimes H\to H, \eta\colon \mathbb{Q}\to H$) satisfying the usual axioms of an algebra (set $1 := \eta(1)$).
- A co-algebra structure ($\Delta\colon H\to H\otimes H, \varepsilon\colon H\to\mathbb{Q}$) satisfying the "dual" axioms and compatible with the algebra structure in the sense that Δ and ε are morphisms of algebras.
- An "antipode" $S\colon H\to H$ which is an anti-homomorphism of both the algebra structure and the co-algebra structure, which is "involutive", $S^2=I$, and which is a "convolution inverse" of the identity map:

$$\Delta /\!\!/ (I\otimes S) /\!\!/ m = \varepsilon /\!\!/ \eta = \Delta /\!\!/ (S\otimes I) /\!\!/ m.$$

Remember the interpretations!
m: Stitch strands. η: Insert an empty strand.
Δ: Double a strand. ε: Delete a strand.
S: Reverse a strand. R (below): A crossing.

Definition. An "R-element" for H (related to "a quasi-triangular structure") is an invertible $R\in H\otimes H$ such that $\bar R := R^{-1}$ inverts R also in $H\otimes H^{op}$ and such that

$$(\Delta\otimes I)(R) = R_{23}R_{13} \quad\text{and}\quad (I\otimes\Delta)(R) = R_{12}R_{13},$$
$$(\eta\otimes I)(R) = 1 = (I\otimes\eta)(R),$$
$$(S\otimes I)(R) = R^{-1} = (I\otimes S)(R).$$

The WG Example

Let G be a finite group with identity element 1 and let $WG := \mathbb{Q}\langle W(\alpha,\beta)\colon \alpha,\beta\in G\rangle$. Set

$$\begin{aligned}
W(\alpha,\beta)W(\gamma,\delta) &:= \delta_{\alpha\beta,\beta\gamma}W(\alpha,\beta\delta),\\
\eta(1) &:= \sum_\alpha W(\alpha,1),\\
\Delta W(\alpha,\beta) &:= \sum_\gamma W(\gamma,\beta)\otimes W(\alpha\gamma^{-1},\beta),\\
\varepsilon W(\alpha,\beta) &:= \delta_{\alpha,1},\\
SW(\alpha,\beta) &:= W(\beta^{-1}\alpha^{-1}\beta,\beta^{-1}),\\
R &:= \sum_{\alpha,\beta} W(\alpha,1)\otimes W(\beta,\alpha),\\
\bar R &= \sum_{\alpha,\beta} W(\alpha,1)\otimes W(\beta,\alpha^{-1}).
\end{aligned}$$

Proposition. WG is an IHOP algebra and R is an R-element for it.
Proof. Think about homomorphisms from the fundamental group of the complement of a tangle to G.

An Implementation of WG

```
DeclareGroup[S_k_] := Module[{α, β, e, γs},
  Clear[G, n, g, ι, m, inv];
  G = PermutationCycles /@ (Permutations@Range@k);
  n = Length[G];
  Do[g[α] = e = G[[α]]; ι[e] = α, {α, n} ];
  m[] = ι[Cycles[{}]];
  Do[m[α, β] = ι[g[α]~PermutationProduct~g[β]],
   {α, n}, {β, n} ];
  m[α_] := α; m[α_, β_, γs__] := m[m[α, β], γs];
  Do[inv[α] = ι[InversePermutation[g[α]]], {α, n}]
 ]
```

```
DeclareGroup[S3];
Table[m[i, j], {i, n}, {j, n}] // MatrixForm
```

$$\begin{pmatrix}
1 & 2 & 3 & 4 & 5 & 6\\
2 & 1 & 4 & 3 & 6 & 5\\
3 & 5 & 1 & 6 & 2 & 4\\
4 & 6 & 2 & 5 & 1 & 3\\
5 & 3 & 6 & 1 & 4 & 2\\
6 & 4 & 5 & 2 & 3 & 1
\end{pmatrix}$$

```
Basis[] = {1};
Basis[i_, is___] :=
 Flatten@Table[W_i[α, β] Basis[is], {α, n}, {β, n}]
```

```
Basis[1, 2]
```

```
{W1[1, 1] W2[1, 1], W1[1, 1] W2[1, 2],
 W1[1, 1] W2[1, 3], W1[1, 1] W2[1, 4],
 W1[1, 1] W2[1, 5], W1[1, 1] W2[1, 6],
 W1[1, 1] W2[2, 1], W1[1, 1] W2[2, 2],
 W1[1, 1] W2[2, 3], ⋯ 1278 ⋯ , W1[6, 6] W2[5, 4],
 W1[6, 6] W2[5, 5], W1[6, 6] W2[5, 6],
 W1[6, 6] W2[6, 1], W1[6, 6] W2[6, 2],
 W1[6, 6] W2[6, 3], W1[6, 6] W2[6, 4],
 W1[6, 6] W2[6, 5], W1[6, 6] W2[6, 6]}
```

large output | show less | show more | show all | set size limit...

```
m_{i_,j_→k_}[ℰ_] :=
  Expand[ℰ] /. W_i[α_, β_] W_j[γ_, δ_] :>
    If[m[α, β] == m[β, γ], W_k[α, m[β, δ]], 0];
η_{i_}[ℰ_] := Expand[ℰ Sum[W_i[α, m[]], {α, n}]];
```

```
Δ_{i_→j_,k_}[ℰ_] :=
  Expand[
   ℰ /. W_i[α_, β_] :>
     Sum[W_j[γ, β] W_k[m[α, inv[γ]], β], {γ, n}]];
ε_{i_}[ℰ_] :=
  Expand[ℰ /. W_i[α_, β_] :> If[α == m[], 1, 0]];
```

```
S_{i_}[ℰ_] :=
  Expand[
   ℰ /. W_i[α_, β_] :> W_i[m[inv[β], inv[α], β],
      inv[β]]]];
```

```
R_{i_,j_} := Sum[W_i[α, m[]] W_j[β, α], {α, n}, {β, n}];
R̄_{i_,j_} := Sum[W_i[α, m[]] W_j[β, inv@α], {α, n},
   {β, n}];
```

```
b = Basis[1, 2, 3];
(b // m_{1,2→1} // m_{1,3→1}) == (b // m_{2,3→2} // m_{1,2→1})
```

True

```
b = Basis[1]; (b // η_2 // m_{1,2→1}) == b == (b // η_2 // m_{1,2→1})
```

True

```
b = Basis[1];
(b // Δ_{1→1,2} // Δ_{2→2,3}) == (b // Δ_{1→1,3} // Δ_{1→1,2})
```

True

```
b = Basis[1]; (b // Δ_{1→1,2} // ε_2) == b == (b // Δ_{1→2,1} // ε_2)
```

True

```
b = Basis[1, 2];
(b // ε_1 // ε_2) == (b // m_{1,2→1} // ε_1)
```

True

```
b = Basis[1, 3];
(b // Δ_{1→1,2} // Δ_{3→3,4} // m_{1,3→1} // m_{2,4→2}) ==
  (b // m_{1,3→1} // Δ_{1→1,2})
```

True

```
b = Basis[1]; (b // S_1 // S_1) == b
```

True

```
b = Basis[1]; (b // Δ_{1→1,2} // S_2 // m_{1,2→1}) ==
  (b // ε_1 // η_1) == (b // Δ_{1→1,2} // S_1 // m_{1,2→1})
```

True

```
(R_{1,2} R̄_{3,4} // m_{1,3→1} // m_{2,4→2}) == (1 // η_1 // η_2) ==
  (R_{1,2} R̄_{3,4} // m_{1,3→1} // m_{4,2→2})
```

True

```
(R_{1,2} R_{4,3} R_{5,6} // m_{1,4→1} // m_{2,5→2} // m_{3,6→3}) ==
  (R_{2,3} R_{1,4} R_{5,6} // m_{1,5→1} // m_{2,6→2} // m_{3,4→3})
```

True

```
{(R_{1,3} // Δ_{1→1,2}) == (R_{2,3} R_{1,4} // m_{3,4→3}),
  (R_{1,2} // Δ_{2→2,3}) == (R_{0,2} R_{1,3} // m_{0,1→1})}
```

{True, True}

```
{(R_{1,2} // ε_1) == (1 // η_2), (R_{1,2} // ε_2) == (1 // η_1)}
```

{True, True}

```
(R_{1,2} // S_1) == R̄_{1,2} == (R_{1,2} // S_2)
```

True

Does R1 hold?

```
{R_{1,2} // m_{1,2→1}, 1 // η_1}
```

```
{W_1[1, 1] + W_1[2, 2] + W_1[3, 3] + W_1[4, 4] +
  W_1[5, 5] + W_1[6, 6], W_1[1, 1] + W_1[2, 1] +
  W_1[3, 1] + W_1[4, 1] + W_1[5, 1] + W_1[6, 1]}
```

```
Ks = {PD[X[1, 4, 2, 5], X[3, 6, 4, 1], X[5, 2, 6, 3]],
   PD[X[4, 2, 5, 1], X[8, 6, 1, 5], X[6, 3, 7, 4],
    X[2, 7, 3, 8]],
   PD[X[1, 6, 2, 7], X[3, 8, 4, 9], X[5, 10, 6, 1],
    X[7, 2, 8, 3], X[9, 4, 10, 5]],
   PD[X[1, 4, 2, 5], X[3, 8, 4, 9], X[5, 10, 6, 1],
    X[9, 6, 10, 7], X[7, 2, 8, 3]]};
Z[pd_PD] := Module[{z},
  z =
   Expand[Times @@ pd /.
     x : X[i_, j_, k_, l_] :>
      If[PositiveQ@x, R_{l,i}, R̄_{j,i}]];
  Do[z = z // m_{1,k→1}, {k, 2 Length@pd}];
  z]
```

```
Table[K → Echo[Timing[Z[K]]], {K, Ks}]
```

```
{1.01563, W_1[1, 1] + 3 W_1[2, 2] +
  3 W_1[3, 3] + W_1[4, 1] + W_1[5, 1] + 3 W_1[6, 6]}
```

$Aborted